NetLogo Hill Climbing Challenge

Basic Description

Your task is to create a breed of turtles that is able — individually or collectively, but with limited
communication abilities — to find the highest point in an unknown terrain. That terrain will be
smooth, but not necessarily convex or unimodal — i.e. there might be many hills that contain the
high spots in their local neighborhoods, but there's no guarantee that the peak of a given hill is
the highest point in the entire terrain.

Time Limits

The challenge will run for an unspecified number of ticks. However, midway through the
running time, turtles will be notified of this milestone via a mark-halftime-<breeds>
procedure. If you want your turtles to change their behaviors after half the time has
elapsed, it's your responsibility to implement that logic in this procedure.

Shortly before time expires, a mark-warning-<breeds> procedure will be invoked for
all turtles. The amount of time remaining will not be specified; however, it will be
enough time for a turtle to move in a straight line from any point in the world to any other
point in the world — but probably not much more than that.

When time expires, only those turtles within a distance of 1 of the highest point on the
terrain will survive; all others will be killed. There is no limit to the number of turtles that
can stand on any given patch, including the patch representing the highest point.

Turtle Abilities and Constraints

You may define a single breed, or multiple breeds; if the latter, the initial combined
population size will be the same as if you had defined a single breed, but the proportions
of the total for multiple breeds will be random.

Every tick, each turtle will be asked to perform the move-<breeds> procedure; you
must include this procedure in your code. This is the only procedure in which your turtles
are allowed to move.

Turtles may not move a distance greater than 1 in any single tick. (Please remember that a
move from the center of a patch to the center of a diagonally adjacent patch is a distance
greater than 1.)

Turtles may not examine any aspect of the terrain at a distance greater than 1. For
example, patch-here, patch-ahead, patch-right-and-ahead, patch-left-
and-ahead, and patch-at-heading-and-distance may all be used by turtles to
refer to nearby patches, as long as the specified distance is less than or equal to 1.
patch-at may also be used, but with more care (since, for example, specifying 1 for dx
and 1 for dy will result in a distance of +/2).

Turtles may remember any number of patches they've visited, but if you want turtles to
use some attributes of such patches (e.g. height, pxcor, pycor) when the patches are
more than 1 step away, the turtles will need to store that information as well. The only
thing turtles are allowed to do with distant patches is face them, use the direction to the
patch given by towards, and get the distance to them. (The implication is that once a
patch is visited, the turtle has a perfect sense of direction to return to the patch, and
knows how far away it is. Anything else must be remembered in custom turtle attributes.)

Turtles may communicate with other turtles from the same breed, or from any of the
breeds defined in your code. This communication may consist of reading other turtles'
attributes, asking other turtles to do some work, etc.

For best results, use breed-specific commands and reporters for communication between
turtles.

Turtles may not communicate with turtles defined in another participant's code.

Turtle communication can only take place between turtles on the same patch — i.e. using
some selection from turtles-here, <breeds>-here, etc..

All standard built-in turtle attributes (who, size, shape, color, breed, heading,
xcor, ycor, etc.) may be read by your code; only heading, xcor, and ycor may be
modified (directly or indirectly) by your code, using the movement primitives.

Every turtle will also have a heal th attribute; this may be read — but not modified — by
your code. Some hazards (below) decrease the value of this attribute; when a turtles'
health falls to 0 or lower, it dies.

You may define and use additional attributes for your turtles, using a <breeds>-own
statement. Your code may freely modify these, as long as no turtle either reads (via of) or
modifies (using ask) the attributes of a turtle that isn't on the same patch. If initialization
of these attributes is needed, your code should include a setup-<breeds> procedure,
which will be called automatically for each turtle. (Don't forget: All uninitialized
attributes have a default value of 0; that may not be appropriate for some attributes,
especially if they're intended to store lists, agent references, or agent sets.)

At the start of play, some number of turtles (the same number for each participant in a
heat) will be placed on the terrain randomly; no turtle will be placed on a hazard patch
(see below).

Turtles may read the standard built-in patch attributes (color, pxcor, pycor, etc.), but
aren't allowed to modify any of them.

Each patch will have a height attribute, which may be read — but not modified — by
your code.

Your code may must not define any custom patch attributes — i.e. you're not allowed to
have a patches-own statement in your code.

The only predefined global variable (besides built-in constants, such as the symbolic



color names, the base-colors list, p1i, e, ctc.) that may be read by the turtles is ticks.
(Of course, turtles aren't allowed to update the clock using tick or tick-advance.) In
particular, turtles aren't allowed to read wor1d-height, world-width, min-pxcor,
min-pycor, max-pxcor, and max-pycor.

* Since the reporters random-xcor, random-ycor, random-pxcor, and random-
pycor give information (particularly when used multiple times) about the world
dimensions, these reporters may not be used by turtles in your code.

* Turtles may not ask patches, ties, or other turtles to perform any action which would
indirectly violate the above rules.

Terrain and Hazards

* The topology of the terrain will be a box; no horizontal or vertical wrapping will be used.
* The actual size of the terrain will not be published in advance.

* All of the normal terrain (including the hilltops) will be colored some shade of brown. (If
needed, use the shade-of? primitive to test for shades of a color.) However, there will
also be hazards on the terrain:

* Radioactive pits — These are red patches; a turtle landing on one of these will die
instantly. Generally, they will placed as individual patches, but random placement
may result in multiple adjacent pits.

* Tar pits — These are green patches, which slow down travel for turtles passing
over them. A turtle landing on one of these will not be able to leave that patch
until a randomly determined number of ticks (between 10 and 20) has elapsed.
These will also be placed individually and randomly.

* Barbed-wire fence — Gray patches which inflict damage, reducing the health of
turtles that land on them. These will be placed in strings of connected patches, but
no more than 10 in a row.

* Turtle-pult — These violet patches are catapults which will fling a turtle to a
random spot on the terrain (but not directly on a hazard patch). This can be good
or bad, in terms of a turtle's location, but they definitely damage a turtle's health,
and will require some number of ticks (between 10 and 20).

The quantitative effects of fences and turtle-pults on turtle health will be somewhat
random, but will not be so great as to cause death with fewer than 5 incidents.



Winning the Game

The preliminary round will consist of heats including 2-5 players, and will be won by the
player with the most surviving turtles — i.e. those that are at the highest point of the
terrain when time expires. If time (and the number of participants) permits, the
preliminary round will be done in round-robin style, so that each player has the
opportunity to compete multiple times.

The final round will include the highest 2-5 scorers (in terms of heats won) from the
preliminary round; there will be 3-5 heats in the final round, with the scores averaged
across the heats to determine the winner.

Clarification and Judging

Requests for clarification of the rules should be directed to instructors and RAs in
advance. Those same people will serve as judges of the event, to resolve any
disagreements or unforeseen ambiguities in the rules, and they will have the last say.
Attempts at bribery are allowed, but aren't guaranteed to be effective.

;; For testing: Change the following line to match the name of your source file.
__includes ["example-turtles.nls"]

extensions [sound]

globals
[
  _breeds
  _breed-colors
  _highest-patch
  _warning-ticks
  _halftime?
  _warning?
  _radiation-death-toll
  _speed-death-toll
  _damage-death-toll
  _failure-death-toll
  _turtle-size
  _turtle-health
  _birds-eye-height
  _gravitational-constant
  _base-theta
  _max-distance
  _peak-safe-radius
  _time-limit
  _color-names
]

turtles-own
[
  health
  _stuck-patch
  _unstuck-ticks
  _last-xy
  _turtle-pult-origin
  _turtle-pult-destination
  _turtle-pult-steps
  _turtle-pult-step
]

patches-own
[
  height
  _fence-length
]

to setup
  clear-all
  setup-globals
  setup-terrain
  setup-fences
  setup-hazards
  setup-all-breeds
  setup-output
  setup-plot
  update-plot
end

;; For testing: Change the first line of this procedure to match the plural name 
;; (without quotes) of your turtle breed. If you have more than one breed, 
;; include all of them in the list.

to setup-globals
  set _breeds (list example-turtles) 
  set _breed-colors [cyan magenta orange blue pink turquoise yellow]
  set _halftime? false
  set _warning? false
  set _max-distance (sqrt (world-height ^ 2 + world-width ^ 2))
  set _warning-ticks (time-limit - 100 * ceiling (_max-distance / 100))
  set _turtle-size 5
  set _turtle-health 100
  set _birds-eye-height 400
  set _gravitational-constant 10
  set _base-theta (2 * atan (_turtle-size / 2) _birds-eye-height)
  set _peak-safe-radius 10
  set _time-limit time-limit
  set _color-names ["gray" "red" "orange" "brown" "yellow" "green" "lime" 
    "turquoise" "cyan" "sky" "blue" "violet" "magenta" "pink"]
end

to setup-terrain
  ask n-of 10 patches
  [
    set height (1000 + random 1000)
  ]
  ask n-of 10 patches
  [
    set height (-1000 + random 500)
  ]
  repeat 40
  [
    diffuse height 0.5
  ]
  set _highest-patch (max-one-of patches [height])
  ask _highest-patch
  [
    set height (height + 500)
  ]
  repeat 20
  [
    diffuse height 0.5
  ]
  let max-height ([height] of _highest-patch)
  let min-height (min [height] of patches)
  ask patches
  [
    set pcolor scale-color brown height min-height max-height
  ]
end

to setup-fences 
  ask patches
  [
    if ((random-float 1) < fence-density)
    [
      set pcolor gray
      set _fence-length 1
    ]
  ]
  while [any? patches with [pcolor = gray and _fence-length < 5]]
  [
    let fences patches with [pcolor = gray]
    let fence-to-tear-down (min-one-of fences [_fence-length])
    let fence-to-extend (one-of (fences with [_fence-length < 10 and (count neighbors4 with [(pcolor != gray) and ((count neighbors with [pcolor = gray]) <= 2)]) >= 2]))
    ask one-of ([neighbors4 with [(pcolor != gray) and ((count neighbors with [pcolor = gray]) <= 2)]] of fence-to-extend)
    [
      set pcolor gray
    ]
    ask fence-to-extend 
    [
      update-fence-length (_fence-length + 1)
    ]
    ask fence-to-tear-down
    [
      set pcolor (mean [pcolor] of (neighbors4 with [pcolor != gray]))
      set _fence-length 0
    ]
  ]
end

to update-fence-length [value]
  set _fence-length value
  ask (neighbors4 with [pcolor = gray and _fence-length < value])
  [
    update-fence-length value
  ]
end

to setup-hazards
  ask (patches with [(pcolor != gray) and (distance _highest-patch > _peak-safe-radius)])
  [
    let selector (random-float 1)
    let normal-patch? true
    ifelse (normal-patch? and selector < radioactive-pit-density)
    [
      set pcolor red
      set normal-patch? false
    ]
    [
      set selector (selector - radioactive-pit-density)
    ]
    ifelse (normal-patch? and selector < tar-pit-density)
    [
      set pcolor green
      set normal-patch? false
    ]
    [
      set selector (selector - tar-pit-density)
    ]
    ifelse (normal-patch? and selector < turtle-pult-density)
    [
      set pcolor violet
      set normal-patch? false
    ]
    [
      set selector (selector - tar-pit-density)
    ]
  ]
end

to setup-all-breeds
  let breed-colors (sublist _breed-colors 0 length _breeds)
  (foreach _breeds breed-colors
  [
    create-turtles population-per-breed 
    [
      set breed ?1
      set shape "turtle"
      set color ?2
      set size _turtle-size
      set health _turtle-health
      setxy random-xcor random-ycor
      while [not shade-of? brown pcolor]
      [
        setxy random-xcor random-ycor
      ]
      carefully
      [
        run (word "setup-" ?1)
      ] 
      [
        log-turtle-error error-message
      ]
      set _last-xy (list xcor ycor)
      set _stuck-patch nobody
      set _turtle-pult-origin nobody
      set _turtle-pult-destination nobody
    ]
  ]) 
end

to setup-output
  let breed-colors (sublist _breed-colors 0 length _breeds)
  (foreach _breeds breed-colors
  [
    let color-index (floor (?2 / 10))
    let color-name (item color-index _color-names)
    output-print (word ?1 " (" color-name ")")
  ])
end

to setup-plot
  set-current-plot "Populations"
  set-plot-y-range 0 population-per-breed
  set-plot-x-range 0 (time-limit + 10)
end

to go
  if (ticks > _time-limit)
  [
    stop
  ]
  if (ticks = _time-limit)
  [
    ask turtles with [distance _highest-patch > 1]
    [
      set _failure-death-toll (_failure-death-toll + 1)
      croak false
    ]
    show-current-stats
    update-plot
    stop
  ]
  ask turtles
  [
    ifelse (is-agent? _turtle-pult-destination)
    [
      travel-by-turtle-pult
    ]
    [
      ifelse (is-agent? _stuck-patch)
      [
        struggle-in-tar-pit
      ]
      [
        carefully 
        [
          run (word "move-" breed)
        ] 
        [
          log-turtle-error error-message
        ]
        if (pcolor = red)
        [
          fall-in-radiation-pit
        ]
        if (distancexy first (_last-xy) last (_last-xy) > 1.00001)
        [
          get-speeding-ticket
        ]
        if (pcolor = green)
        [
          fall-in-tar-pit
        ]
        if (pcolor = violet)
        [
          spring-turtle-pult
        ]
        if (pcolor = gray)
        [
          cross-barbed-wire-fence
        ]
      ]
    ]
    if (health <= 0)
    [
      set _damage-death-toll (_damage-death-toll + 1)
      croak true
    ]
    set _last-xy (list xcor ycor)
  ]
  tick
  if (ticks >= time-limit / 2 and not _halftime?)
  [
    ask turtles
    [
      carefully
      [
        run (word "mark-halftime-" breed)
      ]
      [
        log-turtle-error error-message
      ]
    ]
    set _halftime? true
  ]
  if (ticks >= _warning-ticks  and not _warning?)
  [
    ask turtles
    [
      carefully
      [
        run (word "mark-warning-" breed)
      ]
      [
        log-turtle-error error-message
      ]
    ]
    set _warning? true
  ]
  if (((_time-limit - ticks) mod 10) = 0)
  [
    show-current-stats
  ]
  update-plot
end

to spring-turtle-pult
  set _turtle-pult-origin patch-here
  set _turtle-pult-destination (one-of patches)
  while [member? ([pcolor] of _turtle-pult-destination) (list red green violet)]
  [
    set _turtle-pult-destination (one-of patches)
  ]
  set _turtle-pult-steps (max list 10 ceiling ((20 * distance _turtle-pult-destination) / _max-distance))
  set _turtle-pult-step 0
  if (sound?)
  [
    sound:play-sound "Boing.wav"
  ]
end

to travel-by-turtle-pult
  set _turtle-pult-step (_turtle-pult-step + 1)
  move-to _turtle-pult-origin
  let travel-distance (distance _turtle-pult-destination)
  face _turtle-pult-destination
  forward (travel-distance * _turtle-pult-step / _turtle-pult-steps)
  let climb-height (_gravitational-constant * _turtle-pult-step * (_turtle-pult-steps - _turtle-pult-step) / 2)
  let theta (2 * atan (_turtle-size / 2) (max list 1 (_birds-eye-height - climb-height)))
  set size (_turtle-size * theta / _base-theta)
  if (patch-here = _turtle-pult-destination)
  [
    move-to _turtle-pult-destination
    set _turtle-pult-destination nobody
    set _turtle-pult-origin nobody
    set size _turtle-size
    set health (health - _turtle-pult-steps)
  ]
end

to get-speeding-ticket
  set _speed-death-toll (_speed-death-toll + 1)
  if (sound?)
  [
    sound:play-sound "Siren.wav"
  ]
  croak false
end

to fall-in-radiation-pit
  set _radiation-death-toll (_radiation-death-toll + 1)
  if (sound?)
  [
    sound:play-sound "Zap.wav"
  ]
  croak false
end

to cross-barbed-wire-fence
  set health (health - 10 - random 11)
  if (sound?)
  [
    sound:play-sound "Pain.wav"
  ]
end

to fall-in-tar-pit
  set _stuck-patch patch-here
  set _unstuck-ticks (ticks + 10 + random 11)
  if (sound?)
  [
    sound:play-sound "Splash.wav"
  ]
end

to struggle-in-tar-pit
  ; set health (health - 1)
  if (_unstuck-ticks <= ticks)
  [
    set _stuck-patch nobody
  ]
end

to croak [noisily?]
  if (noisily? and sound?)
  [
    sound:play-sound "Scream.wav"
  ]
  set shape "skull and crossbones"
  stamp
  die
end

to log-turtle-error [message]
  show (word breed ": " message)
end

to show-current-stats
  let breed-colors (sublist _breed-colors 0 length _breeds)
  let unsorted-results (map [(list (word ?1 " (" (item (floor (?2 / 10)) _color-names) "): " (count ?1)) count ?1)] _breeds breed-colors)
  let sorted-results (sort-by [last ?1 > last ?2] unsorted-results)
  clear-output
  foreach sorted-results
  [
    output-print (first ?)
  ]
end

to update-plot
  let breed-colors (sublist _breed-colors 0 length _breeds)
  (foreach _breeds breed-colors
  [
    create-temporary-plot-pen (word ?1)
    set-plot-pen-color ?2
    plot (count ?1)
  ])
end

WHAT IS IT?
-----------
This section could give a general understanding of what the model is trying to show or explain.


HOW IT WORKS
------------
This section could explain what rules the agents use to create the overall behavior of the model.


HOW TO USE IT
-------------
This section could explain how to use the model, including a description of each of the items in the interface tab.


THINGS TO NOTICE
----------------
This section could give some ideas of things for the user to notice while running the model.


THINGS TO TRY
-------------
This section could give some ideas of things for the user to try to do (move sliders, switches, etc.) with the model.


EXTENDING THE MODEL
-------------------
This section could give some ideas of things to add or change in the procedures tab to make the model more complicated, detailed, accurate, etc.


NETLOGO FEATURES
----------------
This section could point out any especially interesting or unusual features of NetLogo that the model makes use of, particularly in the Procedures tab.  It might also point out places where workarounds were needed because of missing features.


RELATED MODELS
--------------
This section could give the names of models in the NetLogo Models Library or elsewhere which are of related interest.


CREDITS AND REFERENCES
----------------------
This section could contain a reference to the model's URL on the web if it has one, as well as any other necessary credits or references.

Polygon -7500403 true true 15 120 150 15 285 120
Line -16777216 false 30 120 270 120

leaf
false
0
Polygon -7500403 true true 150 210 135 195 120 210 60 210 30 195 60 180 60 165 15 135 30 120 15 105 40 104 45 90 60 90 90 105 105 120 120 120 105 60 120 60 135 30 150 15 165 30 180 60 195 60 180 120 195 120 210 105 240 90 255 90 263 104 285 105 270 120 285 135 240 165 240 180 270 195 240 210 180 210 165 195
Polygon -7500403 true true 135 195 135 240 120 255 105 255 105 285 135 285 165 240 165 195

line
true
0
Line -7500403 true 150 0 150 300

line half
true
0
Line -7500403 true 150 0 150 150

pentagon
false
0
Polygon -7500403 true true 150 15 15 120 60 285 240 285 285 120

person
false
0
Circle -7500403 true true 110 5 80
Polygon -7500403 true true 105 90 120 195 90 285 105 300 135 300 150 225 165 300 195 300 210 285 180 195 195 90
Rectangle -7500403 true true 127 79 172 94
Polygon -7500403 true true 195 90 240 150 225 180 165 105
Polygon -7500403 true true 105 90 60 150 75 180 135 105

plant
false
0
Rectangle -7500403 true true 135 90 165 300
Polygon -7500403 true true 135 255 90 210 45 195 75 255 135 285
Polygon -7500403 true true 165 255 210 210 255 195 225 255 165 285
Polygon -7500403 true true 135 180 90 135 45 120 75 180 135 210
Polygon -7500403 true true 165 180 165 210 225 180 255 120 210 135
Polygon -7500403 true true 135 105 90 60 45 45 75 105 135 135
Polygon -7500403 true true 165 105 165 135 225 105 255 45 210 60
Polygon -7500403 true true 135 90 120 45 150 15 180 45 165 90

sheep
false
0
Rectangle -7500403 true true 151 225 180 285
Rectangle -7500403 true true 47 225 75 285
Rectangle -7500403 true true 15 75 210 225
Circle -7500403 true true 135 75 150
Circle -16777216 true false 165 76 116

skull and crossbones
false
0
Circle -7500403 true true 58 -2 182
Circle -7500403 true true 90 105 120
Circle -16777216 true false 105 75 30
Circle -16777216 true false 165 75 30
Line -16777216 false 90 165 210 165
Polygon -7500403 true true 75 285 75 300 255 225 240 210
Polygon -7500403 true true 45 225 60 210 225 285 225 300
Circle -7500403 true true 45 210 30
Circle -7500403 true true 225 210 30
Circle -7500403 true true 210 270 30
Circle -7500403 true true 60 270 30

square
false
0
Rectangle -7500403 true true 30 30 270 270

square 2
false
0
Rectangle -7500403 true true 30 30 270 270
Rectangle -16777216 true false 60 60 240 240

star
false
0
Polygon -7500403 true true 151 1 185 108 298 108 207 175 242 282 151 216 59 282 94 175 3 108 116 108

target
false
0
Circle -7500403 true true 0 0 300
Circle -16777216 true false 30 30 240
Circle -7500403 true true 60 60 180
Circle -16777216 true false 90 90 120
Circle -7500403 true true 120 120 60

tree
false
0
Circle -7500403 true true 118 3 94
Rectangle -6459832 true false 120 195 180 300
Circle -7500403 true true 65 21 108
Circle -7500403 true true 116 41 127
Circle -7500403 true true 45 90 120
Circle -7500403 true true 104 74 152

triangle
false
0
Polygon -7500403 true true 150 30 15 255 285 255

triangle 2
false
0
Polygon -7500403 true true 150 30 15 255 285 255
Polygon -16777216 true false 151 99 225 223 75 224

truck
false
0
Rectangle -7500403 true true 4 45 195 187
Polygon -7500403 true true 296 193 296 150 259 134 244 104 208 104 207 194
Rectangle -1 true false 195 60 195 105
Polygon -16777216 true false 238 112 252 141 219 141 218 112
Circle -16777216 true false 234 174 42
Rectangle -7500403 true true 181 185 214 194
Circle -16777216 true false 144 174 42
Circle -16777216 true false 24 174 42
Circle -7500403 false true 24 174 42
Circle -7500403 false true 144 174 42
Circle -7500403 false true 234 174 42

turtle
true
0
Polygon -10899396 true false 215 204 240 233 246 254 228 266 215 252 193 210
Polygon -10899396 true false 195 90 225 75 245 75 260 89 269 108 261 124 240 105 225 105 210 105
Polygon -10899396 true false 105 90 75 75 55 75 40 89 31 108 39 124 60 105 75 105 90 105
Polygon -10899396 true false 132 85 134 64 107 51 108 17 150 2 192 18 192 52 169 65 172 87
Polygon -10899396 true false 85 204 60 233 54 254 72 266 85 252 107 210
Polygon -7500403 true true 119 75 179 75 209 101 224 135 220 225 175 261 128 261 81 224 74 135 88 99

wheel
false
0
Circle -7500403 true true 3 3 294
Circle -16777216 true false 30 30 240
Line -7500403 true 150 285 150 15
Line -7500403 true 15 150 285 150
Circle -7500403 true true 120 120 60
Line -7500403 true 216 40 79 269
Line -7500403 true 40 84 269 221
Line -7500403 true 40 216 269 79
Line -7500403 true 84 40 221 269

x
false
0
Polygon -7500403 true true 270 75 225 30 30 225 75 270
Polygon -7500403 true true 30 75 75 30 270 225 225 270

default
0.0
-0.2 0 1.0 0.0
0.0 1 1.0 0.0
0.2 0 1.0 0.0
link direction
true
0
Line -7500403 true 150 150 90 180
Line -7500403 true 150 150 210 180

;; Your code must include at least one breed definition, like this one.

breed [example-turtles example-turtle]

;; You can create attributes for your breed, with a <breeds>-own statement.

example-turtles-own
[
  highest-patch
  highest-patch-height
  uphill-mode?
  halftime?
  warning?
]

;; Initialize any turtle attributes for your breed in the setup-<breeds> 
;; procedure. DO NOT INCLUDE ANY MOVEMENT IN THIS PROCEDURE.

to setup-example-turtles
  set highest-patch patch-here
  set highest-patch-height height
  set halftime? false
  set warning? false
end

;; ALL movement for your turtles must be in the move-<breeds> procedure, or in
;; a procedure called (directly or indirectly) by it. If your turtles' 
;; behaviors are dependent on whether the halftime or warning signal has 
;; sounded, your code should test for those conditions in this procedure.

to move-example-turtles
  if (not halftime? and not warning?)
  [
    wander-aimlessly-example-turtles
  ]
  if (halftime?)
  [
    look-for-peaks-example-turtles
  ]
  if warning?
  [
    go-to-highest-example-turtles
  ]
  compare-to-highest-known-example-turtles
  ; Now that I'm done moving, am I missing an opportunity to communicate with other turtles?
end

;; Use this procedure to modify behavioral attributes, if your turtles should 
;; behave differently when time is halfway gone. DO NOT INCLUDE ANY MOVEMENT IN 
;; THIS PROCEDURE.

to mark-halftime-example-turtles
  set halftime? true
  set uphill-mode? true
end

;; Use this procedure to modify behavioral attributes, if your turtles should
;; behave differently when time is almost up. DO NOT INCLUDE ANY MOVEMENT IN
;; THIS PROCEDURE.

to mark-warning-example-turtles
  set halftime? false
  set warning? true
end

;; The following procedures are additional procedures, which aren't called 
;; directly by the test or tournament rig; instead, they're called by 
;; move-example-turtles. To guarantee that these procedures have unique names 
;; (and thus don't conflict with other players' procedures), it's highly 
;; recommended that you use the breed name in these as well.

to wander-aimlessly-example-turtles
  ; Is wiggling the best way to find the highest point?
  right random-float 30
  left random-float 30
  while [not can-move? 1]
  [
    set heading random-float 360
  ]
  if (not member? ([pcolor] of patch-ahead 1) (list red gray)) ; Am I avoiding all hazards?
  [
    forward 1
  ]
end

to look-for-peaks-example-turtles
  ifelse (uphill-mode?)
  [
    let destination (max-one-of neighbors4 [height])
    ifelse (shade-of? brown [pcolor] of destination)
    [
      face destination
      forward 1
    ]
    [
      set heading random-float 360
      forward 1
    ]
    if ((random 300) = 1) ; Am I bored with walking uphill?
    [
      set uphill-mode? false
    ]
  ]
  [
    right random-float 30
    left random-float 30
    if (can-move? 1 and 
      not member? ([pcolor] of patch-ahead 1) (list red gray)) ; Am I avoiding all hazards?
    [
      forward 1
    ]
    if ((random 200) = 1) ; Am I bored with wiggling?
    [
      set uphill-mode? true
    ]
  ]
end

to go-to-highest-example-turtles
  ifelse (highest-patch != patch-here)
  [
    let step-length (min list 1 distance highest-patch)
    face highest-patch
    ifelse (can-move? step-length and 
      not member? ([pcolor] of patch-ahead step-length) (list red gray)) ; Am I avoiding all hazards?
    [
      forward step-length
    ]
    [
      set heading random-float 360 
      forward 1 ; What's wrong with this?
    ]
  ]
  [
    let highest-neighbor (max-one-of neighbors4 [height])
    if (([height] of highest-neighbor) > height)
    [
      face highest-neighbor
      while [not can-move? 1]
      [
        set heading random-float 360
      ]
      forward 1
    ]
  ]
end

to compare-to-highest-known-example-turtles
  ; Compare height of current patch to height of tallest patch encountered so far.
  if (height > highest-patch-height)
  [
    set highest-patch-height height
    set highest-patch patch-here
  ]
end






